/*

Kuru B67
Proposal title: Polls That Matter: Dynamics of Horse Race Polling and Public Evaluation of Poll Reports

HYPOTHESES

Stated-Hyp 1: Motivated reasoning operates when respondents simultaneously encounter conflicting poll results (H1). 

	Test-Hyp 1: Among participants in condition C1, there will be an effect of partisanship on the perceived accuracy of the poll.
	
Stated-Hyp 2: Partisans will differ from one another in the credibility that they associate with polls that have consistent results (H2; C2 and C3 by partisanship).

	Test-Hyp 2: Among participants in condition C2, there will be an effect of partisanship on the perceived accuracy of the poll.
	Test-Hyp 3: Among participants in condition C3, there will be an effect of partisanship on the perceived accuracy of the poll.

Stated-Hyp 3: When polls are inconsistent, however, we expect that individuals will be responsive to methodological quality.

	Test-Hyp 4: Among participants in condition C4, the higher quality poll (Public-Metrics) will be perceived as more accurate than the lower quality poll (KnowPolitics).
	Test-Hyp 5: Among participants in condition C5, the higher quality poll (Public-Metrics) will be perceived as more accurate than the lower quality poll (KnowPolitics).

Stated-Hyp 4: Compared to conditions where poll results are consistent, we expect that inconsistent results will trigger stronger methods-based assessments (H4; C4 and C5 by partisanship vs C2 and C3 by partisanship).

	Test-Hyp 6: The effect of partisanship on the perceived accuracy of the poll will be different in condition C2 vs in condition C4.
	Test-Hyp 7: The effect of partisanship on the perceived accuracy of the poll will be different in condition C3 vs in condition C5.

Stated-Hyp 5: Expert commentary will enhance motivated reasoning for polls with consistent findings (H5; C6 and C7 by partisanship vs C2 and C3 by partisanship). 

	Test-Hyp 8: The effect of partisanship on the perceived accuracy of the poll will be different in condition C2 vs in condition C6.
	Test-Hyp 9: The effect of partisanship on the perceived accuracy of the poll will be different in condition C3 vs in condition C7.

Stated-Hyp 6: In contrast, we expect that expert commentary will reduce motivational biases where polls’ results are inconsistent, because they will serve as an informational corrective (H6; C8 and C9 by partisanship vs C4 and C5 by partisanship).

	Test-Hyp 10: The effect of partisanship on the perceived accuracy of the poll will be different in condition C4 vs in condition C9.
	Test-Hyp 11: The effect of partisanship on the perceived accuracy of the poll will be different in condition C5 vs in condition C8.

Stated-Hyp 7: Partisan commentary, on the other hand, should enhance motivated reasoning (H7; C10 and C11 by partisanship vs C1). 

	Test-Hyp 12: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C10.
	Test-Hyp 13: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C11.

Stated-Hyp 8: If an expert critiques conflicting polls both of which are high quality, we expect the effects of motivational biases will be stronger than in a condition without expert commentary (H8; C12 vs C1)

	Test-Hyp 14: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C12.
	
********************************************************************************

NOTES: There are additional hypotheses relating to moderation effects and an 
additional dependent variable. We decided not to test these hypotheses. 
Specifically,

Stated-Hyp 9: Across all of the conditions, we expect to replicate the earlier studies’ findings that more politically aware respondents will be the most susceptible to motivational biases; we plan to use education and party strength to test this.

Stated-Hyp 10: Aside from the respondents’ perceptions of accuracy, we will also examine whether our manipulations alter respondents’ perceptions of likely election outcomes.
		
*/

clear all
use KuruB67.dta, clear

********************************************************************************

* RECODING

* experiment condition
	tab xtess207, mis
	tab xtess207, nolabel
	gen condition = .
	replace condition = 1 if xtess207 == 1
	replace condition = 1 if xtess207 == 2
	replace condition = 2 if xtess207 == 3
	replace condition = 3 if xtess207 == 4
	replace condition = 4 if xtess207 == 5
	replace condition = 5 if xtess207 == 6
	replace condition = 6 if xtess207 == 7
	replace condition = 7 if xtess207 == 8
	replace condition = 8 if xtess207 == 9
	replace condition = 9 if xtess207 == 10
	replace condition = 10 if xtess207 == 11
	replace condition = 11 if xtess207 == 12
	replace condition = 12 if xtess207 == 13
	label define con_l 1 "C1" 2 "C2" 3 "C3" 4 "C4" 5 "C5" 6 "C6" 7 "C7" 8 "C8" 9 "C9" 10 "C10" 11 "C11" 12 "C12"
	label values condition con_l  
	tab condition xtess207, mis
	
* Partisanship
	tab dov_xparty7, mis
	tab dov_xparty7, nolabel
	gen rep = 1
	replace rep = . if dov_xparty7 == 4
	replace rep = 0 if dov_xparty7 > 4
	label define rep_l 0 "Democrat" 1 "Republican" 
	label values rep rep_l  
	tab rep dov_xparty7, mis
	
* DV: Poll Accuracy
	tab qkuru1, mis
	tab qkuru1, nolabel
	gen pollAcc = qkuru1
	replace pollAcc = . if pollAcc == -1
	tab pollAcc, mis

********************************************************************************

* ANALYSIS

* Test-Hyp 1: Among participants in condition C1, there will be an effect of partisanship on the perceived accuracy of the poll.

	reg pollAcc i.rep if condition == 1
	// support for H1 (p = .000)
	tess 1.rep, init(KuruB67) 
	
* Test-Hyp 2: Among participants in condition C2, there will be an effect of partisanship on the perceived accuracy of the poll.

	reg pollAcc i.rep if condition == 2
	// no support for H2 (p = .454)
	tess 1.rep
	
* Test-Hyp 3: Among participants in condition C3, there will be an effect of partisanship on the perceived accuracy of the poll.

	reg pollAcc i.rep if condition == 3
	// no support for H3 (p = .117)
	tess 1.rep
	
* Test-Hyp 4: Among participants in condition C4, the higher quality poll (Public-Metrics) will be perceived as more accurate than the lower quality poll (KnowPolitics).

	ttest pollAcc == 4 if condition == 4
	// support for H4 (p = .001)
	tess, ttest
	
* Test-Hyp 5: Among participants in condition C5, the higher quality poll (Public-Metrics) will be perceived as more accurate than the lower quality poll (KnowPolitics).

	ttest pollAcc == 4 if condition == 5
	// support for H5 (p = .002)
	tess, ttest
	
* Test-Hyp 6: The effect of partisanship on the perceived accuracy of the poll will be different in condition C2 vs in condition C4.

	reg pollAcc i.condition##i.rep if condition == 2 | condition == 4
	// no support for H6 (p = .170)
	tess 4.condition#1.rep
	
* Test-Hyp 7: The effect of partisanship on the perceived accuracy of the poll will be different in condition C3 vs in condition C5.

	reg pollAcc i.condition##i.rep if condition == 3 | condition == 5
	// no support for H7 (p = .460)
	tess 5.condition#1.rep
	
* Test-Hyp 8: The effect of partisanship on the perceived accuracy of the poll will be different in condition C2 vs in condition C6.

	reg pollAcc i.condition##i.rep if condition == 2 | condition == 6
	// no support for H8 (p = .565)
	tess 6.condition#1.rep
	
* Test-Hyp 9: The effect of partisanship on the perceived accuracy of the poll will be different in condition C3 vs in condition C7.

	reg pollAcc i.condition##i.rep if condition == 3 | condition == 7
	// no support for H9 (p = .939)
	tess 7.condition#1.rep
	
* Test-Hyp 10: The effect of partisanship on the perceived accuracy of the poll will be different in condition C4 vs in condition C9.

	reg pollAcc i.condition##i.rep if condition == 4 | condition == 9
	// no support for H10 (p = .244)
	tess 9.condition#1.rep
	
* Test-Hyp 11: The effect of partisanship on the perceived accuracy of the poll will be different in condition C5 vs in condition C8.

	reg pollAcc i.condition##i.rep if condition == 5 | condition == 8
	// no support for H11 (p = .754)
	tess 8.condition#1.rep
	
* Test-Hyp 12: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C10.

	reg pollAcc i.condition##i.rep if condition == 1 | condition == 10
	// no support for H12 (p = .586)
	tess 10.condition#1.rep
	
* Test-Hyp 13: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C11.

	reg pollAcc i.condition##i.rep if condition == 1 | condition == 11
	// support for H13 (p = .000)
	tess 11.condition#1.rep
	
	* Note: This test seems problematic because C1 and C11 differ in multiple ways including which poll shows Trump ahead.

* Test-Hyp 14: The effect of partisanship on the perceived accuracy of the poll will be different in condition C1 vs in condition C12.

	reg pollAcc i.condition##i.rep if condition == 1 | condition == 12
	// no support for H14 (p = .381)
	tess 12.condition#1.rep
	